Mecánica de Medios Continuos. Ingeniero Geólogo (curso 3º). 2006-07

LABORATORIO DE MECÁNICA DE MEDIOS CONTINUOS

Segunda Sesión. Análisis de Tensiones

Miércoles 28 de noviembre de 2006 (8:30 - 10:30)

Javier Rodríguez y José M.ª Goicolea

Nombre del alumno:
Documento Nacional de Identidad:
Por favor, guarde este fichero con el nombre <DNI>.mws, siendo <DNI> el número de su Documento Nacional de Identidad.

Resuelva las cuestiones indicadas con asterisco (*)

Problema 1a.1 *

Se considera el estado de tensión plana siguiente en un punto determinado de un medio continuo:

$\sigma := \begin{bmatrix} 0 & \tau & 0 \\ \tau & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$. Se pide:

1) Obtener la tensión sobre los planos paralelos al eje x3 que forman 45º con el eje x2, empleando los procedimientos siguientes.

a) *Equilibrio de cuñas.*

```
> restart:
```

- Primer plano

```
> eq1 :=
  tau[1]*sqrt(2)*cos(Pi/4)-sigma[1]*sqrt(2)*sin(Pi/4)+tau;
  # dirección vertical
```

$$eq1 := \tau_1 - \sigma_1 + \tau$$

```
> eq2 :=
  -tau[1]*sqrt(2)*sin(Pi/4)-sigma[1]*sqrt(2)*cos(Pi/4)+tau
  ; # dirección horizontal
```

$$eq2 := -\tau_1 - \sigma_1 + \tau$$

```
> solve({eq1, eq2}, {sigma[1], tau[1]});
```

$$\{\tau_1 = 0, \sigma_1 = \tau\}$$

- Segundo plano*

```
> eq1 :=
  -tau[2]*cos(Pi/4)*sqrt(2)-sigma[2]*sin(Pi/4)*sqrt(2)-tau
  ; # dirección vertical
```

$$eq1 := -\tau_2 - \sigma_2 - \tau$$

```
> eq2 :=
-tau[2]*sin(Pi/4)*sqrt(2)+sigma[2]*cos(Pi/4)*sqrt(2)+tau
; # dirección horizontal
```

$$eq2 := -\tau_2 + \sigma_2 + \tau$$

```
> solve({eq1, eq2}, {sigma[2], tau[2]});
```

$$\{\tau_2 = 0, \sigma_2 = -\tau\}$$

b) *Fórmulas de tensión.*

```
> restart;
> with(LinearAlgebra):
> sigma := <<0, tau, 0>|<tau, 0, 0>|<0, 0, 0>>;
```

$$\sigma := \begin{bmatrix} 0 & \tau & 0 \\ \tau & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

- Primer plano

```
> n1 := <-cos(Pi/4), -sin(Pi/4), 0>;
```

$$n1 := \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ -\frac{\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

```
> t1 := sigma . n1;
```

$$t1 := \begin{bmatrix} -\frac{\tau\sqrt{2}}{2} \\ -\frac{\tau\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

```
> sigma1 := n1 . t1;
```

$$\sigma1 := \tau$$

```
> m1 := <-sin(Pi/4), cos(Pi/4), 0>;
```

$$m1 := \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

```
> tau1 := m1 . t1 ;
```

$$\tau1 := 0$$

- Segundo plano*

```
> n2 := <cos(Pi/4), -sin(Pi/4), 0>;
```

$$n2 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ -\frac{\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

```
> t2 := sigma . n2;
```

$$t2 := \begin{bmatrix} -\frac{\tau\sqrt{2}}{2} \\ \frac{\tau\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

```
> sigma2 := n2 . t2;
```

$$\sigma 2 := -\tau$$

```
> m2 := <-sin(Pi/4), -cos(Pi/4), 0>:
> tau2 := m2 . t2;
```

$$\tau 2 := 0$$

***c)** Cambio de coordenadas de los ejes (x1, x2) a los definidos por los planos pedidos.*

- Primer plano

Expresión genérica de la matriz de rotación alrededor del eje 3:

```
> A := theta -> <<cos(theta), sin(theta), 0>|<-sin(theta),
  cos(theta), 0>|<0, 0, 1>>;
```

$$A := \theta \to \langle\langle \cos(\theta) \mid \sin(\theta) \mid 0\rangle \mid \langle -\sin(\theta), \cos(\theta), 0\rangle \mid \langle 0, 0, 1\rangle\rangle$$

Ejemplo de uso:

```
> A(beta);
```

$$\begin{bmatrix} \cos(\beta) & -\sin(\beta) & 0 \\ \sin(\beta) & \cos(\beta) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Como en este caso el ángulo es $\frac{1}{4}\pi$ a partir del eje 2, aplicamos la expresión anterior para $\frac{1}{4}\pi + \frac{1}{2}\pi$:

```
> A1 := A(Pi/4+Pi/2);
```

$$A1 := \begin{bmatrix} -\frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} & 0 \\ \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> sigmaA1 := Transpose(A1) . sigma . A1;
```

$$sigmaA1 := \begin{bmatrix} -\tau & 0 & 0 \\ 0 & \tau & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> sigma1 := sigmaA1[2,2];
```

$$\sigma 1 := \tau$$

```
> tau1 := sigmaA1[2,1];
```

$$\tau 1 := 0$$

- Segundo plano

```
> tau2 := Column(sigmaA1,2)[1];
```

$$\tau 2 := 0$$

```
> sigma2 := Column(sigmaA1,2)[2];
```

$$\sigma 2 := \tau$$

2) Obtener las tensiones principales ($\sigma_1, \sigma_2, \sigma_3$) y sus direcciones respectivas ($\boldsymbol{n_1}, \boldsymbol{n_2}, \boldsymbol{n_3}$). Calcular a partir de éstas los máximos y mínimos de la tensión normal y tangencial

```
> restart;
> with(LinearAlgebra):
> sigma := <<0, tau, 0>|<tau, 0, 0>|<0, 0, 0>>;
```

$$\sigma := \begin{bmatrix} 0 & \tau & 0 \\ \tau & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> eig := Eigenvectors(sigma);
  tp := eig[1];
```

$$eig := \begin{bmatrix} 0 \\ \tau \\ -\tau \end{bmatrix}, \begin{bmatrix} 0 & 1 & -1 \\ 0 & 1 & 1 \\ 1 & 0 & 0 \end{bmatrix}$$

$$tp := \begin{bmatrix} 0 \\ \tau \\ -\tau \end{bmatrix}$$

Calculamos las tensiones principales y las ordenamos, para lo que necesitamos asignar un valor numérico a τ:

```
> tpf := subs(tau=1,tp);
```

$$tpf := \begin{bmatrix} 0 \\ 1 \\ -1 \end{bmatrix}$$

El cálculo lo realizamos mediante un sencillo bucle, donde asignamos los índices de tensión principal mayor, menor e intermedia

```
> imax:=1:imed:=1:imin:=1:
  for i from 2 to 3 do
    if (evalf(tpf[i]-tpf[imax]) >0) then imax:=i end if:
```

```
    if (evalf(tpf[i]-tpf[imin])<=0) then imin:=i end if:
  end do;
  for i from 1 to 3 do
    if (imax<>i and imin <>i) then imed:=i end if:
  end do;
> imax,imed,imin;
  Sigma[1]:=tp[imax]; Sigma[2]:=tp[imed];
  Sigma[3]:=tp[imin];
```

$$2, 1, 3$$

$$\Sigma_1 := \tau$$

$$\Sigma_2 := 0$$

$$\Sigma_3 := -\tau$$

Direcciones correspondientes, primero obtenemos la matriz característica

```
> cm := CharacteristicMatrix(sigma,lambda);
```

$$cm := \begin{bmatrix} -\lambda & \tau & 0 \\ \tau & -\lambda & 0 \\ 0 & 0 & -\lambda \end{bmatrix}$$

Ahora las direcciones correspondientes a cada tensión principal

```
> for i from 1 to 3 do
    dp[i] := NullSpace(subs(lambda=Sigma[i],cm))[1]:
    dp[i] := Normalize(dp[i],Euclidean);
  end do:
> dp[1],dp[2],dp[3];
```

$$\begin{bmatrix} \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

Comprobaciones:

```
> for i from 1 to 3 do
    simplify(sigma.dp[i])=simplify(Sigma[i]*dp[i]);
  end do;
```

$$\begin{bmatrix} \frac{\tau\sqrt{2}}{2} \\ \frac{\tau\sqrt{2}}{2} \\ 0 \end{bmatrix} = \begin{bmatrix} \frac{\tau\sqrt{2}}{2} \\ \frac{\tau\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

$$\begin{bmatrix} \frac{\tau\sqrt{2}}{2} \\ -\frac{\tau\sqrt{2}}{2} \\ 0 \end{bmatrix} = \begin{bmatrix} \frac{\tau\sqrt{2}}{2} \\ -\frac{\tau\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

Tension tangencial máxima:

```
> tau_max := 1/2 * (Sigma[1]-Sigma[3]);
```

$$tau\_max := \tau$$

Problema 1a.2 *

El estado de un sólido elástico queda definido por las tensiones normales y tangenciales a = 15 MPa de compresión y b = 10 MPa tangencial vertical positiva en un plano vertical, y$c := 15\,2^{\left(\frac{1}{2}\right)}$ de compresión para un plano a 45º. Por otra parte se trata de un estado de deformación plana. La constante elástica de Poisson vale $\nu := \frac{1}{3}$. Se pide:

1) Obtener las componentes del tensor de tensiones en coordenadas cartesianas.

```
> restart;
> with(LinearAlgebra):
> a := 15;
> b := 10;
> c := a*sqrt(2);
> nu := 1/3;
```

$$a := 15$$
$$b := 10$$
$$c := 15\sqrt{2}$$
$$\nu := \frac{1}{3}$$

Escribimos las componentes de tensión, con $s22$ como incógnita:

```
> sigma := <<-a, -b, 0>|<-b, s22, 0>|<0, 0, nu*(-a+s22)>>;
```

$$\sigma := \begin{bmatrix} -15 & -10 & 0 \\ -10 & s22 & 0 \\ 0 & 0 & -5+\frac{s22}{3} \end{bmatrix}$$

Expresión de la tensión normal en el plano dado:

```
> n := <cos(Pi/4), sin(Pi/4), 0>;
```

$$n := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

```
> n.(sigma.n);
```

$$-\frac{25}{2} + \frac{\sqrt{2}\left(-5\sqrt{2} + \frac{s22\sqrt{2}}{2}\right)}{2}$$

La ecuación es por tanto

```
> ecuacion := n.(sigma.n) = -c;
```

$$ecuacion := -\frac{25}{2} + \frac{\sqrt{2}\left(-5\sqrt{2} + \frac{s22\sqrt{2}}{2}\right)}{2} = -15\sqrt{2}$$

Por lo que la solución resulta

```
> solu := solve(ecuacion, s22);
```

$$solu := 35 - 30\sqrt{2}$$

Sustituimos el valor hallado en la matriz de componentes

```
> sigma := subs(s22=solu,sigma);
```

$$\sigma := \begin{bmatrix} -15 & -10 & 0 \\ -10 & 35 - 30\sqrt{2} & 0 \\ 0 & 0 & \frac{20}{3} - 10\sqrt{2} \end{bmatrix}$$

2) Obtener las tensiones principales y la orientación de las mismas

La función <Eigenvectors> nos da autovalores y autovectores (por columnas de la matriz)

```
> Eigenvectors(sigma);
```

$$\begin{bmatrix} 10 - 15\sqrt{2} + 5\sqrt{47 - 30\sqrt{2}} \\ 10 - 15\sqrt{2} - 5\sqrt{47 - 30\sqrt{2}} \\ \frac{20}{3} - 10\sqrt{2} \end{bmatrix},$$

$$\begin{bmatrix} -\frac{10}{25 - 15\sqrt{2} + 5\sqrt{47 - 30\sqrt{2}}} & -\frac{10}{25 - 15\sqrt{2} - 5\sqrt{47 - 30\sqrt{2}}} & 0 \\ 1 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> tp := %[1];
  dp := %%[2];
```

$$tp := \begin{bmatrix} 10 - 15\sqrt{2} + 5\sqrt{47 - 30\sqrt{2}} \\ 10 - 15\sqrt{2} - 5\sqrt{47 - 30\sqrt{2}} \\ \frac{20}{3} - 10\sqrt{2} \end{bmatrix}$$

$$dp := \begin{bmatrix} -\frac{10}{25 - 15\sqrt{2} + 5\sqrt{47 - 30\sqrt{2}}} & -\frac{10}{25 - 15\sqrt{2} - 5\sqrt{47 - 30\sqrt{2}}} & 0 \\ 1 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

(*) ejercicio a realizar

Datos del problema:

```
> restart:
> sigma[2,2]=-20,sigma[1,2]=10,tau(pi/3)=5*(1+sqrt(3)/2),nu=
  1/6;
```

$$\sigma_{2,2} = -20, \sigma_{1,2} = 10, \tau\left(\frac{\pi}{3}\right) = 5 + \frac{5\sqrt{3}}{2}, \nu = \frac{1}{6}$$

```
> with(LinearAlgebra):
> nu:=1/6:
> sigma :=
  Matrix([[s11,10,0],[10,-20,0],[0,0,nu*(s11-20)]]);
```

$$\sigma := \begin{bmatrix} s11 & 10 & 0 \\ 10 & -20 & 0 \\ 0 & 0 & \frac{s11}{6} - \frac{10}{3} \end{bmatrix}$$

Vectores unitarios normal y tangencial:

```
> n := <cos(Pi/3),sin(Pi/3),0>;
  m := <sin(Pi/3),-cos(Pi/3),0>;
```

$$n := \begin{bmatrix} \frac{1}{2} \\ \frac{\sqrt{3}}{2} \\ 0 \end{bmatrix}$$

$$m := \begin{bmatrix} \frac{\sqrt{3}}{2} \\ \frac{-1}{2} \\ 0 \end{bmatrix}$$

Dato de la tensión tangencial

```
> tt := 5*(1+sqrt(3)/2);
```

$$tt := 5 + \frac{5\sqrt{3}}{2}$$

Planteamos la ecuación y la resolvemos para s11:

```
> ecuacion := m.(sigma.n) = tt;
```

$$ecuacion := \frac{\sqrt{3}\left(\frac{s11}{2} + 5\sqrt{3}\right)}{2} - \frac{5}{2} + 5\sqrt{3} = 5 + \frac{5\sqrt{3}}{2}$$

```
> solu := solve(ecuacion, s11);
```

$$solu := -10$$

Sustituimos en la matriz

```
> sigma := subs(s11=solu,sigma);
```

$$\sigma := \begin{bmatrix} -10 & 10 & 0 \\ 10 & -20 & 0 \\ 0 & 0 & -5 \end{bmatrix}$$

```
> tn := n.(sigma.n);
  tn := simplify(%);
```

$$tn := -\frac{5}{2} + \frac{5\sqrt{3}}{2} + \frac{\sqrt{3}\,(5 - 10\sqrt{3})}{2}$$

$$tn := -\frac{35}{2} + 5\sqrt{3}$$

Problema 1.1 *

```
> restart:
> with(LinearAlgebra):
```

Se define el estado de tensión plana siguiente $\sigma := \begin{bmatrix} -1 & 0 \\ 0 & -3 \end{bmatrix}$:

```
> sigma:=Matrix([[-1,0],[0,-3]]);
```

$$\sigma := \begin{bmatrix} -1 & 0 \\ 0 & -3 \end{bmatrix}$$

Se pide obtener las tensiones normal y tangencial para los planos definidos por las normales n1 (-60º) y n2 (30º) por los siguientes procedimientos:

1.a. Equilibrio de cuñas.

Plano de 30º.

Equilibrio de fuerzas horizontales.

```
> Fh_1:=-1*sin(Pi/6)-sigma1.sin(Pi/6)+tau1*cos(Pi/6)=0;
```

Equilibrio de fuerzas verticales.

```
> Fv_1:=3*cos(Pi/6)+sigma1*cos(Pi/6)+tau1*sin(Pi/6)=0;
```

Resolucion del sistema.

```
> solve({Fh_1,Fv_1},{sigma1,tau1});
```

$$Fh\_1 := -\frac{1}{2} - \frac{\sigma 1}{2} + \frac{\tau 1\sqrt{3}}{2} = 0$$

$$Fv\_1 := \frac{3\sqrt{3}}{2} + \frac{\sigma 1\sqrt{3}}{2} + \frac{\tau 1}{2} = 0$$

$$\{\tau 1 = -\frac{\sqrt{3}}{2}, \sigma 1 = \frac{-5}{2}\}$$

Plano de 60º.

```
> Fh_2:=1*cos(Pi/6)+sigma2*cos(Pi/6)+tau2*sin(Pi/6)=0;
> Fv_2:=3*sin(Pi/6)+sigma2*sin(Pi/6)-tau2*cos(Pi/6)=0;
> solve({Fh_2,Fv_2},{sigma2,tau2});
```

$$Fh\_2 := \frac{\sqrt{3}}{2} + \frac{\sigma 2\sqrt{3}}{2} + \frac{\tau 2}{2} = 0$$

$$Fv\_2 := \frac{3}{2} + \frac{\sigma 2}{2} - \frac{\tau 2\sqrt{3}}{2} = 0$$

$$\{\sigma 2 = \frac{-3}{2}, \tau 2 = \frac{\sqrt{3}}{2}\}$$

1.b. expresiónes algebraicas de la tension a partir de σ

Plano de 30º.

```
> n2:=<-sin(Pi/6),cos(Pi/6)>;
  m2:=<cos(Pi/6),sin(Pi/6)>;
```

$$n2 := \begin{bmatrix} \frac{-1}{2} \\ \frac{\sqrt{3}}{2} \end{bmatrix}$$

$$m2 := \begin{bmatrix} \frac{\sqrt{3}}{2} \\ \frac{1}{2} \end{bmatrix}$$

```
> t2:=sigma.n2;
```

$$t2 := \begin{bmatrix} \frac{1}{2} \\ -\frac{3\sqrt{3}}{2} \end{bmatrix}$$

```
> t2_n:=n2.t2;
```

$$t2\_n := \frac{-5}{2}$$

```
> t2_sigma:=t2_n*n2;
```

$$t2\_sigma := \begin{bmatrix} \frac{5}{4} \\ -\frac{5\sqrt{3}}{4} \end{bmatrix}$$

```
> t2_tau:=t2-t2_sigma;
```

$$t2\_tau := \begin{bmatrix} \frac{-3}{4} \\ -\frac{\sqrt{3}}{4} \end{bmatrix}$$

```
> t2_t := m2.t2;
```

$$t2\_t := -\frac{\sqrt{3}}{2}$$

Plano de 60º.

```
> n1:=<cos(Pi/6),sin(Pi/6)>;
  m1:=<sin(Pi/6),-cos(Pi/6)>;
```

$$n1 := \begin{bmatrix} \frac{\sqrt{3}}{2} \\ \frac{1}{2} \end{bmatrix}$$

$$m1 := \begin{bmatrix} \frac{1}{2} \\ -\frac{\sqrt{3}}{2} \end{bmatrix}$$

```
> t1:=sigma.n1;
```

$$t1 := \begin{bmatrix} -\frac{\sqrt{3}}{2} \\ \frac{-3}{2} \end{bmatrix}$$

```
> t1_n:=n1.t1;
```

$$t1\_n := \frac{-3}{2}$$

```
> t1_sigma:=t1_n*n1;
```

$$t1\_sigma := \begin{bmatrix} -\frac{3\sqrt{3}}{4} \\ \frac{-3}{4} \end{bmatrix}$$

```
> t1_tau:=t1-t1_sigma;
```

$$t1\_tau := \begin{bmatrix} \frac{\sqrt{3}}{4} \\ \frac{-3}{4} \end{bmatrix}$$

```
> t1_t := m1.t1;
```

$$t1\_t := \frac{\sqrt{3}}{2}$$

1.d. Cambio de componentes del tensor para rotación de los ejes coordenados.

Definicion de la matriz de Cambio de Coordenadas.

```
> e´1:=<n1>;
```

$$e´1 := \begin{bmatrix} \frac{\sqrt{3}}{2} \\ \frac{1}{2} \end{bmatrix}$$

```
> e´2:=<n2>;
```

$$e´2 := \begin{bmatrix} \frac{-1}{2} \\ \frac{\sqrt{3}}{2} \end{bmatrix}$$

```
> A:=Matrix([e´1,e´2]);
```

$$A := \begin{bmatrix} \frac{\sqrt{3}}{2} & \frac{-1}{2} \\ \frac{1}{2} & \frac{\sqrt{3}}{2} \end{bmatrix}$$

```
> AT:=Transpose(A);
```

$$AT := \begin{bmatrix} \frac{\sqrt{3}}{2} & \frac{1}{2} \\ \frac{-1}{2} & \frac{\sqrt{3}}{2} \end{bmatrix}$$

```
> sigma_A:=(AT.sigma).A;
```

$$sigma\_A := \begin{bmatrix} \frac{-3}{2} & -\frac{\sqrt{3}}{2} \\ -\frac{\sqrt{3}}{2} & \frac{-5}{2} \end{bmatrix}$$

Comprobacion

```
> sigma_A[1,1]=t1_n;
  sigma_A[2,2]=t2_n;
```

```
sigma_A[1,2]=-t1_t;
sigma_A[2,1]=t2_t;
```

$$\frac{-3}{2} = \frac{-3}{2}$$

$$\frac{-5}{2} = \frac{-5}{2}$$

$$-\frac{\sqrt{3}}{2} = -\frac{\sqrt{3}}{2}$$

$$-\frac{\sqrt{3}}{2} = -\frac{\sqrt{3}}{2}$$

Ejercicio a resolver *:

Calcular las tensiones normal y tangencial para planos que formen 45º y 135º con el eje x1

Plano de 45º.

```
> n2:=<-sin(Pi/4),cos(Pi/4)>;
  m2:=<cos(Pi/4),sin(Pi/4)>;
```

$$n2 := \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \end{bmatrix}$$

$$m2 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \end{bmatrix}$$

```
> t2:=sigma.n2;
```

$$t2 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ -\frac{3\sqrt{2}}{2} \end{bmatrix}$$

```
> t2_n:=n2.t2;
```

$$t2\_n := -2$$

```
> t2_sigma:=t2_n*n2;
```

$$t2\_sigma := \begin{bmatrix} \sqrt{2} \\ -\sqrt{2} \end{bmatrix}$$

```
> t2_tau:=t2-t2_sigma;
```

$$t2\_tau := \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ -\frac{\sqrt{2}}{2} \end{bmatrix}$$

```
> t2_t := m2.t2;
```

$$t2\_t := -1$$

Plano de 60º.

```
> n1:=<cos(Pi/4),sin(Pi/4)>;
  m1:=<sin(Pi/4),-cos(Pi/4)>;
```

$$n1 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \end{bmatrix}$$

$$m1 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ -\frac{\sqrt{2}}{2} \end{bmatrix}$$

```
> t1:=sigma.n1;
```

$$t1 := \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ -\frac{3\sqrt{2}}{2} \end{bmatrix}$$

```
> t1_n:=n1.t1;
```

$$t1\_n := -2$$

```
> t1_sigma:=t1_n*n1;
```

$$t1\_sigma := \begin{bmatrix} -\sqrt{2} \\ -\sqrt{2} \end{bmatrix}$$

```
> t1_tau:=t1-t1_sigma;
```

$$t1\_tau := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ -\frac{\sqrt{2}}{2} \end{bmatrix}$$

```
> t1_t := m1.t1;
```

$$t1\_t := 1$$

Problema 1a.4

1. Calcular las tensiones en el estado III.

```
> restart:
> with(LinearAlgebra):
```

Algebraicamente a partir del tensor de tensiones.

Tensiones en el estado I.

```
> sigma_I:=Matrix([[1,2*sqrt(3)],[2*sqrt(3),-3]]);
```

$$sigma\_I := \begin{bmatrix} 1 & 2\sqrt{3} \\ 2\sqrt{3} & -3 \end{bmatrix}$$

Tensiones en el estado II.

Primero expresamos el tensor de tensiones en los ejes rotados x' e y', despues efectuamos un cambio de base para expresarlos en los ejes coordenados x e y.

```
> sigmap_II:=Matrix([[-3,0],[0,1]]);
```

$$sigmap\_II := \begin{bmatrix} -3 & 0 \\ 0 & 1 \end{bmatrix}$$

```
> e´1:=<cos(Pi/6),sin(Pi/6)>;
```

$$e´1 := \begin{bmatrix} \frac{\sqrt{3}}{2} \\ \frac{1}{2} \end{bmatrix}$$

```
> e´2:=<-sin(Pi/6),cos(Pi/6)>;
```

$$e´2 := \begin{bmatrix} \frac{-1}{2} \\ \frac{\sqrt{3}}{2} \end{bmatrix}$$

```
> A:=Matrix([e´1,e´2]);
```

$$A := \begin{bmatrix} \frac{\sqrt{3}}{2} & \frac{-1}{2} \\ \frac{1}{2} & \frac{\sqrt{3}}{2} \end{bmatrix}$$

```
> AT:=Transpose(A);
```

$$AT := \begin{bmatrix} \frac{\sqrt{3}}{2} & \frac{1}{2} \\ \frac{-1}{2} & \frac{\sqrt{3}}{2} \end{bmatrix}$$

Ahora hacemos el cambio inverso para "desrotar" los ejes

```
> sigma_II:=A.sigmap_II.AT;
```

$$sigma\_II := \begin{bmatrix} -2 & -\sqrt{3} \\ -\sqrt{3} & 0 \end{bmatrix}$$

Tensor de tensiones en el estado III.

```
> sigma_III:=sigma_I+sigma_II;
```

$$sigma\_III := \begin{bmatrix} -1 & \sqrt{3} \\ \sqrt{3} & -3 \end{bmatrix}$$

4.2. Tensiones principales e invariantes.

```
> ev := Eigenvalues(sigma_III);
```

$$ev := \begin{bmatrix} 0 \\ -4 \end{bmatrix}$$

```
> tp:=Eigenvectors(sigma_III)[2];
  tp1 := Column(tp,1);
  tp2 := Column(tp,2);
```

$$tp := \begin{bmatrix} -\frac{\sqrt{3}}{3} & \sqrt{3} \\ 1 & 1 \end{bmatrix}$$

$$tp1 := \begin{bmatrix} -\frac{\sqrt{3}}{3} \\ 1 \end{bmatrix}$$

$$tp2 := \begin{bmatrix} \sqrt{3} \\ 1 \end{bmatrix}$$

Comprobacion

```
> sigma_III . tp1 = ev[1] * tp1;
  sigma_III . tp2 = ev[2] * tp2;
```

$$\begin{bmatrix} \frac{4\sqrt{3}}{3} \\ -4 \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} = \begin{bmatrix} -4\sqrt{3} \\ -4 \end{bmatrix}$$

Para el calculo de los invariantes, suponemos tension plana (=0 normal al plano)

Construimos la matriz 3x3

```
> sigma := Matrix(3,3,sigma_III);
```

$$\sigma := \begin{bmatrix} -1 & \sqrt{3} & 0 \\ \sqrt{3} & -3 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> I1:=Trace(sigma);
> I2:=1/2*((Trace(sigma))^2-Trace((sigma)^2));
> I3:=0;
```

$$I1 := -4$$

$$I2 := 0$$

$$I3 := 0$$

Comprobacion

```
> cp := -lambda^3+I1*lambda^2-I2*lambda**2+I3=0;
  solve(cp,lambda);
```

$$cp := -\lambda^3 - 4\,\lambda^2 = 0$$

$$-4, 0, 0$$

4.3. Parte esferica y desviadora.

```
> sigma_m:=1/3*(Trace(sigma));
```

$$sigma\_m := \frac{-4}{3}$$

```
> sigma_desv:=sigma-IdentityMatrix(3)*sigma_m;
```

$$sigma\_desv := \begin{bmatrix} \frac{1}{3} & \sqrt{3} & 0 \\ \sqrt{3} & \frac{-5}{3} & 0 \\ 0 & 0 & \frac{4}{3} \end{bmatrix}$$

Comprobación

```
> traza_sigma_desv:=Trace(sigma_desv);
```

$$traza\_sigma\_desv := 0$$

Problema 1.4 *

(Ver enunciado entregado).

1) Tensiones que actúan en los planos del estado III.

```
> restart;
> with(LinearAlgebra):
> sigmaI := Matrix([[-2, 1, 0], [1, 5, 0], [0, 0, 0]]);
```

$$sigmaI := \begin{bmatrix} -2 & 1 & 0 \\ 1 & 5 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> sigmaIIp := Matrix([[3, 0, 0], [0, 1, 0], [0, 0, 0]]);
```

$$sigmaIIp := \begin{bmatrix} 3 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> fA := x -> <<cos(x), sin(x), 0>|<-sin(x), cos(x), 0>|<0,
  0, 1>>;
```

$$fA := x \rightarrow \langle\langle \cos(x) \mid \sin(x) \mid 0\rangle \mid \langle -\sin(x), \cos(x), 0\rangle \mid \langle 0, 0, 1\rangle\rangle$$

```
> A := fA(Pi/4+Pi/2);
```

$$A := \begin{bmatrix} -\frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} & 0 \\ \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> sigmaII := Transpose(A) . sigmaIIp . A;
```

$$sigmaII := \begin{bmatrix} 2 & 1 & 0 \\ 1 & 2 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> sigmaIII := sigmaI + sigmaII;
```

$$sigmaIII := \begin{bmatrix} 0 & 2 & 0 \\ 2 & 7 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

2) Determinar las direcciones principales y los invariantes.

```
> Eigenvalues(sigmaIII);
```

$$\begin{bmatrix} 0 \\ \frac{7}{2} + \frac{\sqrt{65}}{2} \\ \frac{7}{2} - \frac{\sqrt{65}}{2} \end{bmatrix}$$

```
> I1 := Trace(sigmaIII);
```

$$I1 := 7$$

```
> I2 := 1/2*(Trace(sigmaIII)^2 - Trace(sigmaIII^2));
```

$$I2 := -4$$

```
> I3 := Determinant(sigmaIII);
```

$$I3 := 0$$

3) Determinar las partes esférica y desviadora

```
> sigmaIIIesf := 1/3*Trace(sigmaIII)*Matrix([[1, 0, 0], [0, 
  1, 0], [0, 0, 1]]);
```

$$sigmaIIIesf := \begin{bmatrix} \frac{7}{3} & 0 & 0 \\ 0 & \frac{7}{3} & 0 \\ 0 & 0 & \frac{7}{3} \end{bmatrix}$$

```
> sigmaIIIdev := evalm(sigmaIII - sigmaIIIesf);
```

$$sigmaIIIdev := \begin{bmatrix} \frac{-7}{3} & 2 & 0 \\ 2 & \frac{14}{3} & 0 \\ 0 & 0 & \frac{-7}{3} \end{bmatrix}$$

4.1. Tensiones sobre el plano III.

Algebraicamente a partir del tensor de tensiones.

Tensiones en el Plano I.

```
> sigma_I:=Matrix([[-2,1],[1,5]]);
```

$$sigma\_I := \begin{bmatrix} -2 & 1 \\ 1 & 5 \end{bmatrix}$$

Tensiones en el Plano II.

Primero expresamos el tensor de tensiones en los ejes de los planos perpendiculares sobre los que actuan 1 y 3, y despues efectuamos un cambio de base para expresarlos en los ejes coordenados x e y.

```
> sigma_II:=Matrix([[3,0],[0,1]]);
```

$$sigma\_II := \begin{bmatrix} 3 & 0 \\ 0 & 1 \end{bmatrix}$$

```
> e´1:=<sqrt(2)/2,-sqrt(2)/2>;
```

$$e´1 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ -\frac{\sqrt{2}}{2} \end{bmatrix}$$

```
> e´2:=<sqrt(2)/2,sqrt(2)/2>;
```

$$e´2 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \end{bmatrix}$$

```
> A:=Matrix([e´1,e´2]);
```

$$A := \begin{bmatrix} \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \\ -\frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \end{bmatrix}$$

```
> AT:=Transpose(A);
```

$$AT := \begin{bmatrix} \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \end{bmatrix}$$

```
> sigma_II:=AT.sigma_II.A;
```

$$sigma\_II := \begin{bmatrix} 2 & 1 \\ 1 & 2 \end{bmatrix}$$

Tensor de tensiones en el Plano III.

```
> sigma_III:=sigma_I+sigma_II;
```

$$sigma\_III := \begin{bmatrix} 0 & 2 \\ 2 & 7 \end{bmatrix}$$

4.2. Tensiones principales e invariantes.

```
> Eigenvalues(sigma_III);
```

$$\begin{bmatrix} \frac{7}{2} + \frac{\sqrt{65}}{2} \\ \frac{7}{2} - \frac{\sqrt{65}}{2} \end{bmatrix}$$

```
> tp:=Eigenvectors(sigma_III);
```

$$tp := \begin{bmatrix} \frac{7}{2} + \frac{\sqrt{65}}{2} \\ \frac{7}{2} - \frac{\sqrt{65}}{2} \end{bmatrix}, \begin{bmatrix} \frac{2}{\frac{7}{2} + \frac{\sqrt{65}}{2}} & \frac{2}{\frac{7}{2} - \frac{\sqrt{65}}{2}} \\ 1 & 1 \end{bmatrix}$$

```
> I1:=Trace(sigma_III);
```

$$I1 := 7$$

```
> I2:=1/2*((Trace(sigma_III))^2-Trace((sigma_III)^2));
```

$$I2 := -4$$

```
> I3:=Determinant(sigma_III);
```

$$I3 := -4$$

4.3. Parte esferica y desviadora.

```
> sigma_m:=1/2*(Trace(sigma_III));
```

$$sigma\_m := \frac{7}{2}$$

```
> sigma_desv:=sigma_III-IdentityMatrix(2)*sigma_m;
```

$$sigma\_desv := \begin{bmatrix} \frac{-7}{2} & 2 \\ 2 & \frac{7}{2} \end{bmatrix}$$

```
> traza_sigma_desv:=Trace(sigma_desv);
```

$$traza\_sigma\_desv := 0$$

Problema 1.a5

```
> restart;
> with(LinearAlgebra):
```

Dado el campo de tensiones para una base ortonormal

```
> sigma := Matrix([[(1-x[1]^2)*x[2]+2/3*x[2]^3,
  -(4-x[2]^2)*x[1], 0],
> [-(4-x[2]^2)*x[1], -1/3*(x[2]^3-12*x[2]), 0],
> [0, 0, (3-x[1]^2)*x[2]]]);
```

$$\sigma := \begin{bmatrix} (1-x_1^2)\,x_2 + \frac{2}{3}x_2^3 & -(4-x_2^2)\,x_1 & 0 \\ -(4-x_2^2)\,x_1 & -\frac{1}{3}x_2^3 + 4\,x_2 & 0 \\ 0 & 0 & (3-x_1^2)\,x_2 \end{bmatrix}$$

Se pide

1) Comprobar que se verifican las ecuaciones de equilibrio en todo punto para fuerzas másicas nulas.

```
> for j from 1 to 3 do
> s := 0:
> for i from 1 to 3 do
> s := s + diff(sigma[i, j], x[i]):
> od;
> print(s);
> od:
```

$$0$$
$$0$$
$$0$$

Otro método:

```
> with(VectorCalculus):
Warning, the names &x, CrossProduct and DotProduct have been rebound
Warning, the assigned names <,> and <|> now have a global binding
Warning, these protected names have been redefined and unprotected: *,
+, ., D, Vector, diff, int, limit, series
> SetCoordinates('cartesian'[x[1],x[2],x[3]]);
```

$$cartesian_{x_1, x_2, x_3}$$

```
> F1 := VectorField(Column(sigma,1));
```

$$F1 := \left((1-x_1^2)\,x_2 + \frac{2}{3}x_2^3\right)\bar{\mathbf{e}}_{\mathbf{x[1]}} - (4-x_2^2)\,x_1\,\bar{\mathbf{e}}_{\mathbf{x[2]}}$$

```
> Divergence(F1);
```

$$0$$

```
> F2 := VectorField(Column(sigma,2));
```

$$F2 := -(4 - x_2^{\,2})\, x_1\, \bar{\mathbf{e}}_{\mathbf{x[1]}} + \left(-\frac{1}{3} x_2^{\,3} + 4\, x_2\right) \bar{\mathbf{e}}_{\mathbf{x[2]}}$$

```
> Divergence(F2);
```

$$0$$

```
> F3 := VectorField(Column(sigma,3));
```

$$F3 := (3 - x_1^{\,2})\, x_2\, \bar{\mathbf{e}}_{\mathbf{x[3]}}$$

```
> Divergence(F3);
```

$$0$$

2) Determinar el vector tensión en el punto P = (2, -1, 6) sobre el plano

$\mathbf{3\, x_1 + 6\, x_2 + 2\, x_3 = 12}$ **:**

```
> n := Normalize(<3, 6, 2>, Euclidean);
```

$$n := \begin{bmatrix} \frac{3}{7} \\ \frac{6}{7} \\ \frac{2}{7} \end{bmatrix}$$

```
> x[1] := 2; x[2] := -1; x[3] := 6;
```

$$x_1 := 2$$
$$x_2 := -1$$
$$x_3 := 6$$

```
> Map(x->eval(x), sigma . n);
```

$$\left(\frac{-29}{7}\right) \mathbf{e}_{\mathbf{x[1]}} - \frac{40}{7}\, \mathbf{e}_{\mathbf{x[2]}} + \frac{2}{7}\, \mathbf{e}_{\mathbf{x[3]}}$$

```
>
>
```